

```
10 ' #####
20 ' #
30 ' #   FILL SUBROUTINE for CPC 464   #
40 ' #                               #
50 ' #           by Ruud Way           #
60 ' #                               #
70 ' #####
```

```

80 '
90 '-----
100 ' PROGRAM DRAWS CONE + RANDOM SHAPES
110 '-----
120 RANDOMIZE TIME
130 MODE 1
140 INK 0,26:INK 1,13:INK 2,2:INK 3,1
150 BORDER 26:CLG 0
160 PAPER 0:PEN 1
170 ORIGIN 320,130
180 '
190 'shadows
200 MOVE 160,60:DRAW 26,-14,1
210 MOVE 160,60:DRAW 34,28,1
220 MOVE 26,-14:DRAW 2,148,3
230 '
240 'half ellipse
250 ellip=2:n%=24:r%=20
260 s=SIN(-PI/n%)
270 c=COS(-PI/n%)
280 odx=r%
290 ody=0
300 MOVE ellip*r%,0
310 FOR t%=1 TO n%
320   dx=odx*c-ody*s
330   dy=ody*c+odx*s
340   odx=dx
350   ody=dy
360   IF t%<=7 THEN col%=3 ELSE col%=2
370   DRAW ellip*dx,dy,col%
380 NEXT t%
390 '
400 'cone
410 DRAW 0,150,2
420 MOVE 2,150:DRAW ellip*r%,0,3
430 '
440 'fill in cone and shadows
450 bacol%=0
460 incol%=2:xseed%=-38:yseed%=0 :GOSUB 1080
470 incol%=3:xseed%=38 :yseed%=0 :GOSUB 1080
480 incol%=3:xseed%=6 :yseed%=128:GOSUB 1080
490 incol%=1:xseed%=36 :yseed%=26 :GOSUB 1080
500 '
510 WHILE INKEY$<>"" :WEND
520 LOCATE 1,25:PRINT"press any key to continue";
530 WHILE INKEY$="" :WEND
540 '
550 '
560 'next frame
570 CLG 0
580 INK 0,0:INK 1,26:INK 2,6:INK 3,2
590 ORIGIN 245,125
600 BORDER 0
610 '
620 'box
630 w%=150
640 MOVE 0,0
650 DRAW w%,0,1
660 DRAW w%,w%,1
670 DRAW 0,w%,1
680 DRAW 0,0,1
690 '
700 'random outlines
710 MOVE 10,10
720 FOR i%=1 TO 10
730   DRAW w%*RND,w%*RND,1
740 NEXT i%
750 DRAW 10,10,1
760 '
770 'fill in random generated shape
780 xseed%=w%\2:yseed%=w%\2
790 incol%=2:bacol%=0
800 GOSUB 1080
810 '
820 'fill box, if possible
830 incol%=3:bacol%=0
840 xseed%=w%-2:yseed%=2 :GOSUB 1080
850 xseed%=w%-2:yseed%=w%-2:GOSUB 1080
860 xseed%=2 :yseed%=2 :GOSUB 1080
870 xseed%=2 :yseed%=w%-2:GOSUB 1080
880 '
890 t=TIME
900 WHILE TIME-t<300:WEND

```



```

910 CLG 0:GOTO 640
920 'end of program
930 '
940 '-----
950 '          FILL SUBROUTINE
960 '-----
970 'enter: seedcoordinates      xseed%,yseed%
980 '          backgroundcolour  bacol%
990 '          fillcolour        incol%
1000 '
1010 'real variables regain their value after returning from FILL
1020 '
1030 'changes for mode 0 & 2:
1040 'mode 2: left=1, all x-stepvalues=1
1050 '( for instance line 1140: x2=x2+1 )
1060 'mode 0: left=4, all x-stepvalues=4
1070 '
1080 DEFINT a-z
1090 DIM stack(1,100)
1100 ptr=0
1110 hole=0
1120 '
1130 x2=xseed%
1140 WHILE TEST(x2,yseed%)=bacol%:x2=x2+2:WEND
1150 '
1160 x1=xseed%
1170 IF TEST(x1,yseed%)=bacol% THEN WHILE TEST(x1-2,yseed%)=bacol%:x1=x1-2:WEND
1180 '
1190 y1.old=yseed%
1200 y2.old=yseed%
1210 '
1220 FOR x=x1 TO x2 STEP 2
1230 '
1240   y1=yseed%
1250   IF TEST(x,y1)=bacol% THEN WHILE TEST(x,y1-2)=bacol%:y1=y1-2:WEND
1260 '
1270   y2=yseed%
1280   IF TEST(x,y2)=bacol% THEN WHILE TEST(x,y2+2)=bacol%:y2=y2+2:WEND
1290 '
1300   IF x<x2 THEN MOVE x,y1:DRAW x,y2,incol%
1310 '
1320   IF ABS(y1-y1.old)<=2 AND y2-y1>2 AND x<>x1 THEN 1400 'ELSE
1330     IF y1.old>y1 THEN left=2:stp=2 ELSE left=0:stp=-2
1340     FOR y=y1 TO y1.old STEP stp
1350       tst=TEST(x-left,y)
1360       IF NOT hole AND tst=bacol% THEN hole=-1:ytemp=y
1370       IF hole AND tst<>bacol% THEN hole=0:stack(0,ptr)=x-left:
         stack(1,ptr)=(ytemp+y-stp)\2:ptr=ptr+1
1380     NEXT y
1390     IF hole THEN hole=0:stack(0,ptr)=x-left:
         stack(1,ptr)=(ytemp+y1.old)\2:ptr=ptr+1
1400 ' FI
1410 '
1420   IF ABS(y2-y2.old)<=2 AND y2-y1>2 AND x<>x1 THEN 1500 'ELSE
1430     IF y2.old<y2 THEN left=2:stp=-2 ELSE left=0:stp=2
1440     FOR y=y2 TO y2.old STEP stp
1450       tst=TEST(x-left,y)
1460       IF NOT hole AND tst=bacol% THEN hole=-1:ytemp=y
1470       IF hole AND tst<>bacol% THEN hole=0:stack(0,ptr)=x-left:
         stack(1,ptr)=(ytemp+y-stp)\2:ptr=ptr+1
1480     NEXT y
1490     IF hole THEN hole=0:stack(0,ptr)=x-left:
         stack(1,ptr)=(ytemp+y2.old)\2:ptr=ptr+1
1500 ' FI
1510 '
1520   y1.old=y1
1530   y2.old=y2
1540 '
1550 NEXT x
1560 '
1570 'pop stack and fill
1580 IF ptr<>0 THEN ptr=ptr-1:xseed%=stack(0,ptr):yseed%=stack(1,ptr):GOTO 1130
1590 '
1600 ERASE stack
1610 DEFREAL a-z
1620 RETURN

```